

AUSLEUCHTEN VON PARABOLSPIEGELN MIT HORNANTENNEN
 FEEDING PARABOLIC DISHES WITH HORN ANTENNAS

by Ulf Hülsenbusch DK2RV

Teil II

Part II

6.	Grenzen für f/D	6.	Limits of f/D
6.2	Bedingung für E-Sektorhorn	6.2	E-sectorhorn performance
6.3	Programmbeschreibung	6.3	Description of program
7.	Beispiel	7.	Example
7.1	Spiegelparameter	7.1	Dish parameters
7.2	E-Sektorhorn	7.2	E-sector horn
7.3	Konstruktion	7.3	Construction
7.4	Messergebnisse	7.4	Measurements

6.1 Grenzen für f/D * limits of f/D

D:Es existiert ein $f/D = 0.24 - 0.30$ bei dem der Spiegel optimal mit einem offenen Rundhohlleiter ausgeleuchtet werden kann. Offene Rechteckhohlleiter sollten wegen der sehr unterschiedlichen Öffnungswinkel in E- und H-Ebene nicht verwendet werden. Bei einem E-Sektorhorn liegt das optimale f/D bei $0.24 - 0.45$.

E:For $f/D = 0.24 - 0.30$ a dish can be fed with an open circular waveguide. An open rectangular waveguide should not be used as its beamwidth in E- and H-plane differ too much. The optimum f/D for an E-sector horn is about $0.24 - 0.45$.

6.2 Bedingung für E-Sektorhorn * E-sector horn performance

D:Beim E-Sektorhorn wird die Breitseite a_0 des Hohlleiters beibehalten. Nur die Schmalseite b_0 wird trichterförmig erweitert. Wie bereits erwähnt läßt sich ein E-Sektorhorn wesentlich einfacher realisieren. Welche Bedingungen müssen eingehalten werden:

1. Bei Eindeutigkeit der H_{10} -Welle gilt $0.5 < a_0/\lambda < 1.0$. Mit bekanntem Fokuswinkel ergeben sich für das H-Diagramm (Fig.10b) die Abszissenwerte $0.5 \cdot \sin \theta$ und $1.0 \cdot \sin \theta$, sowie die zugehörigen Schnittpunkte mit der Kurve $t=0$.

2. Die berechnete Randfeldstärke E/E_0 muß zwischen den beiden Feldstärken der Bereichsgrenzen liegen.

E:An E-sector horn maintains the width a_0 of the rectangular waveguide. Only the height b_0 is extended. As mentioned before, an E-sector horn can be performed much easier than a pyramidal horn. Which conditions must be satisfied?

1. The H_{10} (TE_{10})-Mode requires $0.5 < a_0/\lambda < 1.0$. These values and the focus angle leads to the abscissa limits $0.5 \cdot \sin$ and $1.0 \cdot \sin$ in the H-diagram (Fig.10b) and the corresponding points of field strength for the curve $t=0$.

2. The required field strength E/E_0 must lie within these points.

6.3 BASIC-Programm * BASIC-program

D:Das Programm besteht aus 2 Teilen. Der 1. Teil berechnet alle Spiegelparameter. Eingabe ist der Durchmesser D und das f/D-Verhältnis oder die Tiefe x. Der zweite Teil druckt Werte für ein Pyramidenhorn. Diese werden iterativ mithilfe einer Lagrange-Interpolation berechnet. Hierzu müssen Stützstellen aus dem E- und H-Diagramm eingegeben werden. Das Programm prüft auch die Möglichkeit eines E-Sektorhorns.

E:The BASIC-program consists of 2 parts. The first part calculates the dish parameters. Input is the diameter D and the f/D-ratio or the depth x. The second part computes practicable dimensions for a pyramidal horn. This is done iteratively by a Lagrange interpolation. A table with sample values from the E- and H-plane can be read from Fig. 10a+b. The program also examines if an E-sector horn is practicable.

7. Beispiel * Example

7.1 Spiegel Parameter * Dish parameters

D:Ein Spiegel dessen Durchmesser D=45 cm und dessen Tiefe x=64 mm beträgt, soll im 9-cm Band mit einem E-Sektorhorn ausgeleuchtet werden. $f/D=D/16/x=0.45$. Nun können alle übrigen Parameter den Diagrammen Fig.2-4 entnommen werden oder mit dem Programm berechnet werden.

E:A 45 cm dish with the depth x=6.4 cm is fed with an E-sector horn in the 9 cm band. $f/D=D/16/x=0.45$. The dish parameters can be seen now from fig.2-4 or can be computed from the program.

Dish parameters:

Frequency	3456 MHz	Edge loss	-2.3 dB
Lambda	86.8 mm	Edge Taper	10.0 dB
Diameter	45.0 cm	Edge Field	7.7 dB
Fokus	20.3 cm	Fokus angle	2*58 Deg
f/D	0.45	Edge E/E0	0.415

7.2 E-Sektor Horn * E-sector horn

D:Aus den Bedingungen für das E-Sektorhorn ist zu sehen, daß für die obere Grenze $a/\lambda = 1$ bzw. $a/\lambda \cdot \sin 58 = 0.83$ die Feldstärke ziemlich genau der geforderten Randfeldstärke von $E/E0=0.415$ entspricht. Die Breitseite $a0$ wird mit 80 mm festgelegt, so daß ein Sicherheitsabstand von 8% zur Bereichsgrenze gewährleistet ist. ($a/\lambda = 0.92$). Die Schmalseite $b0$ des Hohlleiters ist innerhalb der Bedingung von Fig.7 frei wählbar und wird auf 10 mm festgelegt. Nun muß die Seite b des Horns berechnet werden. Hierzu wird zunächst $E/E0=0.415$ in das E-Diagramm Fig.10a als horizontale Linie eingetragen. Die Schnittpunkte für $s=0, 0.1, 0.2$ liegen sehr dicht zusammen. Dies bedeutet, daß mechanische Abweichungen sehr geringe Änderungen in der Feldstärke bewirken. Für die Konstruktion ist es daher von Vorteil mit kleinen s-Werten zu operieren. Für $s=0.1$ wird die Abszisse $b/\lambda \cdot \sin \theta = 0.64 \rightarrow b=66$ mm, $s=0.1 \rightarrow Le=62$ mm. Die Hohlleiterwellenlänge ist mit 223 mm sehr groß durch den Betrieb nahe der Bereichsgrenze $a/\lambda = 1$.

E: From the conditions of the E-sectorhorn we can see, that for the upper limit $a/\lambda = 1$ respectively $a/\lambda \cdot \sin 58 = 0.83$ the field strength is nearly the same as the required $E/E_0 = 0.415$. The broadside $a = a_0$ is determined to 80 mm, so that the upper margin is, 8% away ($a/\lambda = 0.92$). The small side b_0 of the waveguide must be within the limits of Fig.7a and is determined to 10 mm. Now the side b of the horn must be calculated. $E/E_0 = 0.415$ is plotted as a horizontal line into the E-diagram Fig.10a. The intersections for the curves $s = 0, 0.1, 0.2$ are very close together. This means, that mechanical deviations cause almost no change in the electrical characteristic. Thus it is an advantage to operate with small values for s . For $s = 0.1$ the abscissa is $b/\lambda \cdot \sin 58 = 0.64 \rightarrow b = 66 \text{ mm}$
 $s = 0.1 \rightarrow L_e = 62 \text{ mm}$. The operation near the upper limit $a/\lambda = 1$ results in a very high WG-wavelength of 223 mm.




Figure 10a E-plane universal pattern of a rectangular horn






Figure 10b H-plane universal pattern of a rectangular horn

7.3 Konstruktion * Construction

D: Das Horn ist einfach zusammenzubauen. Zunächst werden die Messingbleche zugeschnitten und gebohrt, dann zusammengelötet. Die Einkopplung erfolgt $\lambda_w/4$ vom kalten Ende mit einem Koppelstift.

E: The horn can be easily constructed. First cut the brass sheets and bend it, then solder it up. The coupling to the WG is accomplished with a probe $\lambda_{wg}/4$ from the short end.






7.4 Meßergebnisse * Measurements

D: Horn und Spiegel wurden in E- und H-Ebene vermessen. Die vorausgerechneten Werte für die Feldstärke beim Fokuswinkel liegen etwa um 8% neben den gemessenen Werten. Die Ergebnisse sind in nachfolgenden Diagrammen zusammengestellt.

E: The horn and the dish were measured in the E- and the H-plane. The calculated field strength of the focus angle differs from the measured value about 8%. The results are plotted in the following diagrams.



```

10 REM PROGRAM FOR DISH PARAMETERS AND HORN DIMENSIONS
20 REM ULF HÜLSENBUSCH DK2RV 31.7.86
30 REM LOG = LOGARITHMUS NATURALIS,TRIG.FUNCTIONS IN RADIAN
40 DIM A(12),B(12),S(12),T(12),L(25)
50 PI=4*ATN(1):AD=10:FP=0:CLS
60 PRINT "ENTER DISH PARAMETERS":PRINT
70 INPUT "FREQUENCY/MHz ",F
80 INPUT "DIAMETER/cm ",D
90 INPUT "f/D (ENTER 0 IF UNKNOWN )",FD
100 IF FD<>0 GOTO 120
110 INPUT "DEPTH X/cm ",X:FD=D/X/16
120 L1=300000!/F:REM LAMBDA IN FREE SPACE
130 FO=FD*D:REM FOKUS IN CM
140 INPUT "EDGE TAPER/dB (TYPICAL 10 dB) ",AD
150 TO=2*ATN(1/4/FD):ST=SIN(TO)
160 AR=COS(TO/2):AR=20*LOG(AR*AR)/LOG(10)
170 G=20*LOG(10*PI*D/L1)/LOG(10)-3
180 E=10^-((AD+AR)/20)
190 LPRINT "      DISH PARAMETERS"
200 LPRINT "      *****"
210 LPRINT "      FREQUENCY/MHz: ";TAB(25);USING "#####";F
220 LPRINT "      LAMBDA/mm: ";TAB(25);USING "####.#";L1
230 LPRINT "      DIAMETER/cm: ";TAB(25);USING "####.#";D
240 LPRINT "      FOKUS/CM: ";TAB(25);USING "####.#";FO
250 LPRINT "      GAIN/dBi (n=50%): ";TAB(25);USING "###.#";G
260 LPRINT "      EDGE LOSS/dB: ";TAB(25);USING "###.#";AR
270 LPRINT "      EDGE TAPER/dB: ";TAB(25);USING "###.#";AD
280 LPRINT "      EDGE FIELD/dB";TAB(25);USING "###.#";AD+AR
290 LPRINT "      FOKUSANGLE/DEGR: ";TAB(25);"2*";USING "###.#";TO/PI*360/2
300 LPRINT "      f/D: ";TAB(25);USING "#.###";FD
310 LPRINT "      EDGE E/Eo: ";TAB(25);USING "#.###";E
320 LPRINT
330 IF FD<.24 THEN 380
340 IF FD <.3 THEN LPRINT "      DISH GOOD FOR OPEN CIRCULAR WAVEGUIDE"
350 LPRINT "      DISH GOOD FOR CONICAL HORN"
360 IF FD<.45 THEN LPRINT "      DISH GOOD FOR E-SECTOR HORN"
370 LPRINT "      DISH GOOD FOR PYRAMIDAL HORN"
380 CLS
390 PRINT "DISH PARAMETERS (D)"
400 PRINT "HORN DIMENSIONS (H)"
410 INPUT "END      (E)",A$
420 IF A$="E" THEN STOP
430 IF A$="D" GOTO 50
440 REM
450 REM ITERATIVE CONSTRUCTION OF PYRAMIDAL HORN
460 REM *****
470 CLS
480 INPUT "ENTER NUMBER OF SAMPLES IN THE E-PLANE ",MB
490 FOR I=1 TO MB
500 INPUT "B*sin(TO)/LAMBDA ",B(I)
510 IF FP=0 THEN B(I)=B(I)*L1/ST
520 INPUT "S ",S(I):NEXT I
530 NEXT I
540 INPUT "ENTER NUMBER OF SAMPLES IN THE H-PLANE ",MA
550 FOR I=1 TO MA
560 INPUT "A*sin(TO)/LAMBDA ",A(I)
570 IF FP=0 THEN A(I)=A(I)*L1/ST
580 INPUT "T ",T(I):NEXT I:PRINT
590 GOSUB 1550:REM E-SECTOR HORN
600 CLS:PRINT "ENTER WAVEGUIDE DIMENSIONS FOR PYRAMIDAL HORN"
610 PRINT
620 INPUT "WAVEGUIDE WIDTH a/mm ",AH

```

```

630 IF AH<L1 AND AH>L1/2 GOTO 660
640 PRINT "WAVEGUIDE WIDTH a OUT OF RANGE",CHR$(7)
650 GOTO 620
660 INPUT "WAVEGUIDE HIGHT b/mm ",BH
670 IF BH/AH < .5*L1/AH GOTO 700
680 PRINT "WAVEGUIDE HIGHT b OUT OF RANGE",CHR$(7)
690 GOTO 660
700 PRINT
710 AO=AH/L1:BO=BH/L1
720 LPRINT:LPRINT
730 LZ=L1/SQR(1-L1*L1/4/AH/AH)
740 LPRINT "      PYRAMIDAL HORN"
750 LPRINT "      *****":LPRINT
760 LPRINT "      WAVEGUIDE WIDTH a = ";AH;" mm"
770 LPRINT "      WAVEGUIDE HIGHT b = ";BH;" mm"
780 LPRINT "      WAVELENGTH IN WG = ";LZ;" mm"
790 LPRINT "      WAVELENGTH SPACE = ";L1;" mm"
800 LPRINT:LPRINT
810 LPRINT "      Lh/mm          A/mm          Le/mm          B/mm          L/mm          ERR
OR %"
820 LPRINT
830 GOSUB 1400
840 DS=(S(MB)-S0)/25:REM STEPWIDTH =MAX. S-RANGE/25
850 DT=(T(MA)-T0)/25:REM STEPWIDTH =MAX. T-RANGE/25
860 FOR I=1 TO 25
870 T=DT*I:GOSUB 1060:REM LAGRANGE-A
880 LH=A*A/8/L1/T:LA=(A-AH)
890 FOR J=1 TO 25
900 S=DS*J:IF I=1 THEN GOSUB 1180:REM B-LAGRANGE SUBROUTINE FOR I=1
910 LE=L(J)*L(J)/8/L1/S
920 LA=LA*SQR(LE/A*LE/A+.25):REM A-CONDITION
930 LB=(L(J)-BH)
940 WB=LH/L(J):LB=LB*SQR(WB*WB+.25):REM B-CONDITION
950 FE=ABS(1-LA/LB):REM REL. ERROR FOR MECHANICAL CONDITION
960 IF FE<.05 THEN GOSUB 1310:REM PRINT DIMENSIONS FOR ERROR<5%
970 NEXT J
980 NEXT I
990 LPRINT:LPRINT"      MAX. ERROR FOR THIS ITERATION WAS ";FO*100;" %"
1000 IF FP=0 THEN PRINT "NO PYRAMIDAL HORN ";CHR$(7)
1010 CLS:PRINT "ENDE (E) DISH (D) HORN (H) NEW WAVEGUIDE (W)";CHR$(7):INPUT A$
1020 IF A$="D" THEN 50
1030 IF A$="H" THEN 450
1040 IF A$="W" THEN 600
1050 STOP
1060 REM SUBROUTINE FOR LAGRANGE-INTERPOLATION A=A(T)
1070 A=0
1080 FOR IA=1 TO MA
1090 NA=1:ZA=1
1100 FOR JA=1 TO MA
1110 IF IA=JA THEN GOTO 1140
1120 ZA=ZA*(T-T(JA)):REM ZAEHLER
1130 NA=NA*(T(IA)-T(JA)):REM NENNER
1140 NEXT JA
1150 A=A+A(IA)*ZA/NA
1160 NEXT IA
1170 RETURN
1180 REM SUBROUTINE FOR LAGRANGE-INTERPOLATION B=B(S)
1190 B=0
1200 FOR IB=1 TO MB
1210 NB=1:ZB=1
1220 FOR JB=1 TO MB
1230 IF IB=JB THEN GOTO 1260
1240 ZB=ZB*(S-S(JB)):REM ZAEHLER

```

```

1250 NB=NB*(S(IB)-S(JB)):REM NENNER
1260 NEXT JB
1270 B=B+B(IB)*ZB/NB
1280 NEXT IB
1290 L(J)=B
1300 RETURN
1310 REM PRINT MECHANICAL DIMENSIONS OF PYRAMIDAL HORN
1320 FP=1
1330 LPRINT USING "#####.##";LH;
1340 LPRINT USING "#####.##";A;
1350 LPRINT USING "#####.##";LE;
1360 LPRINT USING "#####.##";L(J);
1370 LPRINT USING "#####.##";(LA+LB)/2;
1380 LPRINT USING "#####.##";FE*100
1390 RETURN
1400 REM BERECHNEN DER BEREICHSGRENZEN
1410 IF A(1)>A0 THEN TO=T(0):GOTO 1470
1420 FOR I=1 TO MA
1430 IF A(I) > A0 GOTO 1450
1440 NEXT I
1450 A2=A(I):A1=A(I-1):T2=T(I):T1=T(I-1)
1460 TO=(A0-A1)*(T2-T1)/(A2-A1)+T1
1470 IF B(1) > B0 THEN SO=S(0):RETURN
1480 FOR I=1 TO MB
1490 IF B(I)>B0 GOTO 1510
1500 NEXT I
1510 B2=B(I):B1=B(I-1)
1520 S2=S(I):S1=S(I-1)
1530 SO=(B0-B1)*(S2-S1)/(B2-B1)+S1
1540 RETURN
1550 REM SUBROUTINE FOR E-SECTOR HORN
1560 T=0
1570 GOSUB 1060:REM A-LAGRANGE
1580 IF A<.5*L1 OR A>L1 GOTO 1840
1590 LS=L1/SQR(1-L1/A*L1/A/4)
1600 DS=.01
1610 CLS
1620 PRINT "INPUT FOR WAVEGUIDE OF E-SECTORHORN":PRINT
1630 PRINT "ENTER WAVEGUIDE HIGHT b < ";A/2;" mm ";:INPUT BH
1640 LPRINT:LPRINT
1650 LPRINT "      E-SECTOR HORN"
1660 LPRINT "      *****"
1670 LPRINT
1680 LPRINT "      WAVEGUIDE HIGHT b   = ";BH;" mm"
1690 LPRINT "      WAVEGUIDE WIDTH a   = ";A;" mm"
1700 LPRINT "      WAVE LENGTH IN WG   = ";LS;" mm"
1710 LPRINT
1720 LPRINT "      B/mm      Le/mm      L/mm"
1730 LPRINT
1740 FOR I=1 TO 10
1750 S=DS*I:GOSUB 1180:REM B-LAGRANGE
1760 LE=B*B/8/L1/S
1770 WB=LE/B:LB=(B-BH)*SQR(WB*WB+.25)
1780 IF LB>2*L1 OR B>2*L1 GOTO 1820
1790 LPRINT USING "#####.##";B;
1800 LPRINT USING "#####.##";LE;
1810 LPRINT USING "#####.##";(LA+LB)/2
1820 NEXT I
1830 LPRINT:LPRINT:LPRINT:RETURN
1840 LPRINT "      NO E-SECTOR "
1850 RETURN

```